5 Verhalten (Behaviors) und ihre Befehle

5.1 Was sind Verhalten?

Seit der Director-Version 6.0 heißen Drehbuch-Scripte Verhalten. Die Grundidee ist jedoch geblieben: ein Skript lässt sich im Drehbuch auf ein Sprite (Spriteskript), oder in den Skript-Kanal des Frames (Frameskript) ziehen. Drehbuchskripte bzw. Verhalten verarbeiten Ereignisse, die direkt auf die Objekte auf der Bühne ein​wirken, nämlich die vom Benutzer erzeugten, wie Maus-Klicks und Tastatureingaben, oder Director-generierte, wie der Wechsel von einem Frame zu einem anderen. Einige neue Ereignisse kamen mit der Einführung der Verhalten hinzu, so z.B. on prepareSprite. Außerdem wurde die Beschränkung aufgehoben, pro Sprite nur ein Skript zuordnen zu können. Es können nun beliebig viele Skripte auf ein Sprite gezogen werden, es darf aber weiterhin nur ein Frameskript pro Frame geben.

Verhalten stellen um Ereignishandler erweiterte Objektskripte dar. Ein Verhalten kann genauso wie ein Parent-Skript eigene Eigenschaften und Methoden definieren, alle objektorientierten Befehle gelten uneingeschränkt. Um das Erzeugen der Instanzen kümmert sich Director automatisch: trifft der Abspielkopf auf ein Sprite, dem ein Verhalten hinzugefügt wurde, wird automatisch eine neue Instanz generiert, verlässt der Abspielkopf das Sprite, wird die Instanz automatisch verworfen.

Die herausragende Eigenschaft der Verhalten ist jedoch, dass sie sich perfekt in die visuelle Arbeitsweise des Autorentools Director einbinden lassen. Ohne großen Aufwand lassen sich Verhaltenseigenschaften über ein konfigurierbares Dialogfenster setzen und verändern.
Um sich einen Überblick über die Möglichkeiten von Verhalten zu verschaffen, ist es empfehlenswert, sich mit den Verhalten vertraut zu machen, die standardmäßig mit Director ausgeliefert werden. Diese sind seit Version 7 im Menü Window unter Library Palette zu finden. Um sie in eigene Filme einzubauen genügt es, diese direkt in das Drehbuch oder auf die Bühne zu ziehen. Director sorgt automatisch dafür, dass sie in die eigene Beset​zung kopiert werden, und zwar jeweils nur einmal. Optional kann man sie auch selbst in die Besetzung ziehen, dabei wird jedes Mal eine neue Kopie angelegt. Die Verhalten werden mit Namen und einem Icon angezeigt. Verweilt der Mauszeiger länger auf einem Icon, wird ein Hilfe-Text eingeblendet, der eine kurze Beschreibung liefert. Es lohnt sich, mit diesen Verhalten zu experimentieren, um ein Gefühl für die vielfältigen Möglichkei​ten zu bekommen, die man durch das Kombinieren verschiedener Verhalten erhält. Die Verhalten liegen alle in ungeschützter Form vor, man kann sich also den Quelltext anzeigen lassen. Auf den ersten Blick wirken sie sehr kompliziert, da sie sehr allgemein gehalten sind. Verhalten selbst zu schreiben ist aber kaum schwieriger als das Erstellen eines herkömmlichen Drehbuchskripts.

Macromedia bietet auf seiner Website eine PDF-Dokument zum Download an, das alle mitgelieferten Verhal​ten beschreibt. Die deutsche Version ist zu finden unter:
http://www.macromedia.com/support/director/how/d7/behaviors_de.html

Die Library Palette stellt übrigens keine unveränderbare Menge von Skripten dar. Man kann leicht seine eige​nen Verhalten hinzufügen, wenn man sie als Bestandteil einer externen Besetzung in das Verzeichnis Libs im Director Installationsverzeichnis kopiert. Die darin enthaltene Ordnerstruktur wird von Director automatisch als Menüstruktur in der Library Palette dargestellt. Eigene Icons werden mit Hilfe des Thumbnail-Menüs in der Eigenschaften-Dialogbox aus der Zwischenablage eingefügt. Ist das Bild der Zwischenablage größer als 43×32 Pixel, wird es automatisch skaliert.

Hinzufügen eines eigenen Verhalten-Icons

Das zentrale Werkzeug zum Verwalten von Verhalten ist der Verhaltensinspektor bzw. Behavior Inspector. Mit ihm lässt sich nachprüfen, wie viele und welche Verhalten einem Sprite zugeordnet sind und welche Werte die Eigenschaften haben. Eine kleine Beschreibung gibt Auskunft wie sich ein Verhalten einsetzen lässt. Außerdem lassen sich einfache Verhalten per Mausklick erstellen.

In der folgenden Grafik werden die Schaltflächen des Verhaltensinspektors beschrieben.

Im Verhaltensinspektor können Verhalten samt ihrer Eigenschaften per „Copy & Paste“ kopiert und einem anderen Sprite zugewiesen werden. Das wiederholte Setzen von Eigenschaften kann so vermieden werden.

Bei Doppelklick auf ein Verhalten in der Besetzung wird per Voreinstellung der Verhaltensinspektor geöffnet. Gerade in der Programmierphase erleichtert es den Arbeitsfluss jedoch erheblich, wenn sich das Skriptfenster direkt öffnet. Dies lässt sich im Menü File-Preferences-Editors (bzw. Datei-Voreinstellungen-Editoren) än​dern, indem als Editor für Verhalten nicht der Verhaltensinspektor, sondern das Skriptfenster ausgewählt wird.

Zusammengefasst bieten Verhalten im Vergleich zu herkömmlichen Drehbuchskripten folgende neue Funktio​nen:

· Verhalten können eigene Eigenschaften definieren.

· Eigenschaften eines Verhaltens können über eine komfortable Eingabemaske geändert werden.

· Einem Sprite können mehrere Verhalten zugewiesen werden.

· Verhalten sind vollwertige Objekte, alle Lingobefehle für Parent-Skripte sind auch auf Verhalten anwend​bar.
5.2 Was können Verhalten?

Macromedia Director ist ein Autorenwerkzeug. Es ist so angelegt, dass Konzeption, Design und Programmie​rung einer einfachen Anwendung von ein und derselben Person ausgeführt werden können. Bei komplexeren Aufgaben ist jedoch üblich, dass die Programmierung und die Montage der Grafiken von verschiedenen Perso​nen ausgeführt werden. Verhalten schaffen 2 Bearbeitungsebenen, um dieser Arbeitsweise gerecht zu werden. Der Programmierer programmiert wie bisher seine Skripte im Skriptfenster. Zusätzlich sorgt er dafür, dass bei der Montage Eingabemasken und Kommentare zur Verfügung stehen. Der Monteur muss sich mit den Skripten nicht mehr befassen – er arbeitet wie gewohnt mit visuellen Arbeitsmitteln.

In der Praxis sieht das so aus, dass zu einem Verhalten in der Besetzung eine Beschreibung vorhanden ist, die über den Einsatz des Verhaltens informiert. Wird ein Verhalten in das Drehbuch gezogen, öffnet sich ein Ein​gabedialog, über den entsprechenden Parameter eingegeben werden können. Für jedes Verhalten, das in das Drehbuch gezogen wird, können andere Werte angegeben werden. Je nach Typ der Eingabe werden die Werte in Textfelder eingegeben oder komfortabel per Pull-Down-Menü und Schieberegler ausgewählt.

In der folgenden Grafik sind die zwei Ebenen für ein Skript dargestellt, dass vom Monteur für das Abspielen von Sounds eingebaut wird. Es muss lediglich der Name des Soundfiles sowie der zu benutzende Soundkanal eingegeben werden. Der Programmierer hat im Verhalten schon die nötigen Befehle zum Abspielen des Sounds implementiert.

[image: image3.wmf]Lingo

-Ebene

Lingo

-Ebene

Montage-Ebene

Montage-Ebene


Für den Benutzer stellt das Verhalten eine „Black Box“ dar. Er muss sich nur um die Eingabeparameter küm​mern und das Verhalten führt eine bestimmte Aktion aus. Was im Verhalten vorgeht, muss nur der Program​mierer wissen. Die Benutzung des Verhaltens hat den Vorteil, dass der Monteur sich mit der Programmierung in Lingo nicht auskennen muss. Dadurch, dass sich die Soundaufrufe nur um wenige Parameter unterscheiden, kann immer das gleiche Skript benutzt werden. Dadurch wird der Code übersichtlich und speichereffizient. Es ist außerdem sehr einfach, bei veränderten Bedingungen dass Soundskript zu überarbeiten. Director fragt daraufhin, ob die schon eingegebenen Werte übernommen werden sollen, oder ob wieder auf die Standardwerte zurückgeschalten werden soll.

Übrigens sollte man aus dieser Arbeitsweise nicht schließen, dass sich das Arbeiten mit Verhalten nur ab einer Teamgröße von 2 Personen aufwärts lohnt. Auch für einen einzelnen Autor ist das Programmieren von Ver​halten empfehlenswert. Der geringe Mehraufwand bei der Implementierung der zusätzlichen Funktionen gleicht sich durch die Trennung von Programmierung und Montage schnell aus. Dazu kommt, dass die Ver​halten meist ohne größere Änderungen in andere Projekte übernommen werden können.

5.3 Das Director Event-Modell

Director bedient sich einer ereignisgesteuerten Metapher: Das Eintreten eines Ereignisses (oder Events) löst eine bestimmte Aktion aus. Ereignisse können vom Benutzer ausgelöst werden, zum Beispiel durch das Drücken einer Taste oder das Bewegen der Maus, oder durch Director selbst, beispielsweise beim Wechsel zu einem anderen Film oder Verlassen des aktuellen Frames. Ein Verhalten besteht in erster Linie aus Ereignis​handlern, die diese Ereignisse verarbeiten. Im Zuge der Einführung von Verhalten hat sich die Zahl der verar​beiteten Ereignissen vergrößert; einerseits sind neue Ereignisse hinzugekommen, wie on beginSprite und on prepareFrame, andererseits sind in Spriteskripten nun Ereignishandler zulässig, die vorher nur im Frameskript abgearbeitet wurden, wie z.B. on exitFrame. Dadurch wird die Beschränkung aufgehoben, dass es nur ein Skript dieser Art pro Frame geben kann, im Gegensatz zum Skriptkanal können einem Sprite beliebig viele Verhalten zugeordnet werden.

5.3.1 Zeitlicher Ablauf von Director-Events

Um Verhalten effizient einzusetzen, ist es sinnvoll, sich klar zu machen, wann Director welches Ereignis aus​löst. Besonderes Augenmerk sollte zwei Aspekten geschenkt werden: wie oft wird ein Ereignis ausgelöst und wird es vor oder nach dem Zeichnen des Sprites auf der Bühne ausgelöst.

Die folgende Tabelle enthält die wichtigsten Director-gesteuerten Ereignisse.

	Ereignishandler
	Skripttypen
	Beschreibung

	on prepareMovie
	Filmskript
	Wird beim Start eines Filmes ausgeführt, und zwar nachdem Darsteller geladen, aber bevor Verhaltensinstanzen erzeugt sind und die Bühne gezeichnet wird.

	on startMovie
	Filmskript
	Wird beim Start eines Filmes ausgeführt, und zwar nachdem die Bühne gezeichnet wird.

	on stopMovie
	Filmskript
	Wird ausgeführt, wenn der Film anhält.

	on beginSprite
	Spriteskript, Frameskript
	Wird ausgeführt, wenn der Abspielkopf auf ein Sprite trifft, das im vorherigen Frame noch nicht existierte.

	on stepFrame
	Objektskript in the actorlist
	Wird ausgeführt, wenn der Abspielkopf in einen neuen Frame wechselt oder updateStage ausgeführt wird, und zwar noch bevor on prepareFrame.

	on prepareFrame
	Spriteskript, Frameskript, Movieskript
	Wird ausgeführt, wenn der Abspielkopf in einen neuen Frame wechselt, und zwar bevor der Frame gezeichnet wird.

	on enterFrame
	Spriteskript, Frameskript, Movieskript
	Wird ausgeführt, wenn der Abspielkopf in einen neuen Frame wechselt, und zwar nachdem der Frame gezeichnet wird.

	on idle
	Frameskript, Filmskript
	Wird ausgeführt, wenn der Film auf das Ende eines Frames wartet, mindestens jedoch 1x pro Frame.

	on exitFrame
	Spriteskript, Frameskript, Movieskript
	Wird ausgeführt, wenn der Abspielkopf einen Frame verlässt.

	on endSprite
	Spriteskript, Frameskript
	Wird ausgeführt, wenn der Abspielkopf ein Sprite verlässt, das im darauffolgenden Frame nicht mehr existiert.


Die folgende Grafik soll den Ablauf noch mal verdeutlichen:

5.3.2 Die Ereigniskette

Benutzer-Ereignisse, wie ein Mausklick oder ein Tastendruck, gehen in Director einen vorgeschriebenen Weg. Zuerst wird das Ereignis an den sogenannten primären Eventhandler weitergegeben, das ist das Skript, das in den Eigenschaften the mouseUpScript, the mouseDownScript, the keyUpscript und the keyDownScript angegeben ist. Wenn das Ereignis in einem Sprite aufgetreten ist, wenn z.B. auf das Sprite geklickt wurde, wird es zuerst an alle Verhalten weitergegeben. Wenn es hier nicht verarbeitet wird, geht es an das Darstellerskript weiter, daraufhin an das Frameskript und dann an das Filmskript. Dabei führt das Vor​handensein eines Eventhandlers in der Regel zum Abbruch der Eventkette, mit Ausnahme von zwei Fällen: der primäre Eventhandler reicht das Event standardmäßig durch, außerdem werden in der Regel alle Handler eines Verhaltens ausgeführt, auch wenn ein Verhalten schon den entsprechenden Eventhandler definiert.

Um die Ereigniskette zu beeinflussen, bietet Lingo die beiden Funktionen pass und stopEvent. Mit pass kann ein Weiterleiten des Ereignisses erzwungen werden, während stopEvent die Ereigniskette abbricht. Wenn einem Sprite mehrere Verhalten zugeordnet sind, kann z.B. das erste Verhalten darüber entscheiden, ob die Ereignisse an die restlichen Sprites weitergegeben werden. Das folgende Skript stellt so ein Verhalten dar. Es sollte an erster Stelle im Verhaltensinspektor stehen. Mit dem Aufruf sprite(x).setActive() wird es dazu veranlasst, alle Mausereignisse an weitere Verhalten des selben Sprites weiterzugeben. Mit dem Aufruf sprite(x).setInactive() schirmt es alle Mausereignisse ab. Darunter liegende Mausevent-Handler werden nicht mehr aufgerufen – das Sprite ist für Mausereignisse deaktiviert.

property mActive

on beginSprite me
  mActive = false
end beginSprite
on setActive me
  -- Mausevents werden fortan durchgereicht

  mActive = true
end setActive

on setInactive me
  -- Mausevents werden fortan gestoppt

  mActive = false
end setInactive

on checkStatus me
  -- stopt das Event, wenn Sprite inaktiv ist

  if not mActive then
    stopEvent
  end if
end checkStatus

on mouseDown me
  me.checkStatus()

end mouseDown
on mouseUp me
  me.checkStatus()

end mouseUp
on mouseUpOutside me
  me.checkStatus()

end mouseUpOutside
on rightMouseUp me
  me.checkStatus()

end rightMouseUp
on mouseEnter me
  me.checkStatus()

end mouseEnter
on mouseLeave me
  me.checkStatus()

end mouseLeave
on mouseWithin me
  me.checkStatus()

end mouseWithin

Der Handler on beginSprite wird dazu benutzt, die Eigenschaften des Verhaltens zu initialisieren. In unserem Fall wird festgesetzt, dass das Sprite zuerst einmal nicht aktiv ist. Es ist zu beachten, dass dieses Ver​halten nur als erstes Verhalten eines Sprites wirksam ist! Die Reihenfolge, mit der Verhalten Sprites zugeord​net werden, ist also entscheidend. Eventuell muss die Reihenfolge mit den Sortieren-Buttons des Verhaltensin​spektors geändert werden.

5.4 Beispiel 8: Der Rechte-Mausbutton-Bugfix

Das abschließende Beispiel ist ein Skript aus der Praxis und soll zeigen, dass sich Eventhandler benutzen las​sen wie gewöhnliche Methoden. Das rightMouseBugFix-Verhalten beseitigt einen Fehler in Director, der auf​tritt, wenn man eine PC-Maus mit zwei Maustasten benutzt. Hält man die linke Taste gedrückt und betätigt kurz die rechte, erhält Director kein mouseUp-Event mehr. Das macht sich z.B. negativ bemerkbar, wenn man ein Objekt per „Drag & Drop“ bewegt und unbeabsichtigt auf die rechte Taste kommt: das Objekt bleibt am Mauszeiger hängen und wird nicht mehr abgesetzt.

Das folgende Skript dient dazu, diesen Fehler nachzuvollziehen. Es ist ein einfaches „Drag & Drop“-Skript, das auf ein Sprite auf der Bühne, z.B. ein Bitmap, gezogen werden sollte.

property mClickOffset, spriteNum
on mouseDown me
  set mClickOffset = the loc of sprite spriteNum - the clickLoc
  (the actorlist).add(me)

end mouseDown
on mouseUp me
  (the actorlist).deleteOne(me)

end mouseUp
on stepframe me
  sprite(spriteNum).loc = the mouseLoc + mClickOffset

end stepframe

Kommt man beim Ziehen der Grafik auf die rechte Maustaste, lässt sich das gezogene Objekt nicht mehr abset​zen.

Dieser Fehler lässt sich leicht und universell mit dem rightMouseBugFixObj-Skript beheben.

-- dieses Objekt ist allen behaviors hinzuzufügen; es verhindert,

-- daß folgender Bug sich negativ auswirkt:

-- hält man unter Windows die linke Maustaste gedrückt

-- und klickt mit der rechten, wird kein Mausup-Event ausgelöst

-- Lösung: jeder rightMouseUp bei gedrückter linker Maustaste wird

-- behandelt wie ein Klick auf den linken Mausbutton
on rightMouseUp me
  if the mouseDown then
    me.mouseUp()

  end if
end rightMouseUp
Wenn die rechte Maustaste losgelassen wird, wird geprüft, ob gerade die linke Taste gedrückt ist. Wenn dies der Fall ist, wird einfach das mouseUp-Skript des Objektes (bzw. der Unterklasse) aufgerufen. Dadurch wird das gezogene Objekt abgesetzt.

Nun muss nur noch in unserem Drag&Drop-Verhalten das rightMouseBugFixObj als ancestor definiert werden.

property ancestor
on beginSprite me
  ancestor = new(script "rightMouseBugFixObj")
end beginSprite
Der Bugfix lässt sich so leicht für sämtliche Verhalten eines Films einsetzen.

5.5 Anzeige von Informationen mit on getBehaviorDescription und on getBehaviorTooltip

Wie jeder anderen Programmcode sollten Verhalten reichlich Kommentartext enthalten. Dieser ist jedoch nur zu sehen, wenn man sich den Text des Verhaltens im Skriptfenster anzeigen lässt. Damit die Funktion eines Verhaltens auch für Nicht-Programmierer ersichtlich wird, lässt sich ein beschreibender Text angeben, der im Verhaltensinspektor erscheint, wenn man das Verhalten markiert. 

Der Handler on getBehaviorDescription macht nichts anderes, als eine Zeichenkette zurückzugeben, die im Beschreibungsfensters des Verhaltensinspektor erscheint. In unserem Sound-Beispiel von vorhin sieht sie z.B. so aus:

on getBehaviorDescription me
  return "Spielt einen Sound ab." & RETURN ¬

  & "Der Sound muss sich im " & QUOTE & "Ton" & QUOTE ¬

  & "-Ordner auf der gleichen Ebene" ¬

  & " befinden wie der aktuelle Film. Die Endung .aif nicht eingeben!"
end getBehaviorDescription
Es gelten die üblichen Regeln für Zeichenketten. Mehrere Zeichenketten können mit dem Verbindungsoperator „&“ zu einer Zeichenkette verbunden werden. Anführungszeichen werden durch das Codewort QUOTE darge​stellt. Zeilenumbrüche können mit dem Codewort RETURN eingefügt werden.

Der Handler on getBehaviorTooltip hat nur Bedeutung  für Verhalten, die in der Library Palette ste​hen. Er hat die gleiche Syntax wie on getBehaviorDescription, mit dem Unterschied, dass der zu​rückgegebene Text in einem kleinen Pop-Up-Fenster angezeigt wird, wenn man mit der Maus auf einem Ver​halten stehen bleibt („Tooltip“). Wie on getBehaviorDescription ist auch dieser Handler optional. Ist er nicht vorhanden, wird der Name des Verhaltens angezeigt.

Tooltip eines Verhaltens aus der Behavior Library

5.6 Eingabe von Parametern mit on getPropertyDescriptionList und on runPropertyDialog
Eine der herausragendsten Eigenschaften von Verhalten ist das Eingeben von Eigenschaftswerten über eine Dialogbox. Wird ein Verhalten in das Drehbuch oder auf die Bühne gezogen bzw. der Parameter-Button im Verhaltensinspektor gedrückt, öffnet sich ein Dialogfenster zur komfortablen Eingabe der individuellen Werte. Diese gelten nur für die aktuelle Instanz, jede neue Instanz erhält eigene Werte. Die Werte werden automatisch mit dem Director-Film abgespeichert.

Zum Erstellen dieses Dialogs dient der Handler on getBehaviorDescriptionList. Dieser Handler ist optional, er muss also nur für die Verhalten implementiert werden, die die Eingabe von Werten erwarten. Er gibt eine Eigenschaftsliste zurück, die für jede zu setzende Eigenschaft des Verhaltens eine Beschreibung über Format und Standardwert sowie die Form des Eingabefeldes enthält. Diese Beschreibung ist wieder eine Eigen​schaftsliste. Da bei mehreren Eigenschaften diese Liste schnell unübersichtlich wird, empfiehlt es sich, das Erstellen der Liste in mehrere Programmzeilen aufzugliedern, wie es im folgenden Skript geschieht:

on getBehaviorDescriptionList
description = [:]

description.addProp(#Eigenschaftsname1, Beschreibungsliste1)

description.addProp(#Eigenschaftsname2, Beschreibungsliste2)
...

return description

end getBehaviorDescriptionList
Die Eigenschaften müssen mit dem property-Schlüsselwort als solche definiert sein, da sie sonst nicht ge​speichert werden. Sie werden als Symbol, also mit vorangestelltem # angegeben, da Lingo sonst versucht, ihren Wert zu interpretieren.

Die Beschreibungsliste, ebenfalls eine Eigenschaftsliste, muss folgende Eigenschaften definieren
:

	#comment: "Kommentartext"
	Der Text, der Auskunft über die einzugebende Eigenschaft gibt.

	#format: #Datentyp
	Der erwartete Datentyp. Zulässige Werte sind: #boolean, #bitmap, #button, #digitalvideo, #field, #filmloop, #float, #frame, #graphic, #ink, #integer, #marker, #member, #movie, #ole, #palette, #picture, #richtext, #script, #shape, #sound, #string, #symbol, #transition, #xtra

	#default: Standardwert
	Dieser Wert ist standardmäßig im Dialogfenster ausgewählt.


Die Definition der Eigenschaft mTon in unserem Soundskript sieht beispielsweise folgendermaßen aus:

on getPropertyDescriptionList
  description = [:]

  description.addProp(#mTon, [ #comment: "Soundname ohne Endung:",  ¬

                               #format:  #string, ¬

                               #default: ""])

  ...

  return description

end getPropertyDescriptionList
Wir legen fest, dass mTon eine Zeichenkette sein soll und als Standardwert nichts im Eingabefeld stehen soll. Das Dialogfenster sieht so aus:

Wichtig ist, dass das Fenster alle eingegebenen Werte auch tatsächlich in dem Format zurückgibt, das für #format eingegeben wurde. Eine nachträgliche Konvertierung, hier z.B. mit der Funktion string() erüb​rigt sich also.
Die Datentypen, die für Format anzugeben sind, lassen sich in zwei Gruppen aufteilen: Darstellertypen, also Datentypen, die in der Besetzung gespeichert werden, und Nicht-Darstellertypen.

Bei der Angabe eines Darstellertypen, z.B. #bitmap, stellt Director automatisch eine Liste aller verfügbaren Darsteller in der Besetzung zusammen und zeigt diese in einem Pull-Down-Menü an. Ein Verhalten, dass einen Darsteller auf der Bühne durch einen anderen austauscht, sobald die Maus über ihm liegt, könnte seine Eigenschaft mRolloverMember wie folgt definieren:

on getPropertyDescriptionList
  description = [:]

  if the currentSpriteNum > 0 then
    memberNumber = sprite(the currentSpriteNum).member.number
    description.addProp(#mRolloverMember,¬

                [#comment: "Rollover-Darsteller:",¬

                 #format: #bitmap,¬

                 #default: member (memberNumber + 1)])

  end if

  return description

end getPropertyDescriptionList
Das Skript benutzt einen ausgefeilteren Standardwert, nämlich immer den Bitmap-Darsteller, der auf den Dar​steller folgt, auf den das Skript gezogen wurde. Wird das Skript in den Spritkanal gezogen, wird gar kein Dia​log angezeigt. (Ab Director 8 ist es möglich zu verhindern, dass Skripte in die falschen Kanäle gezogen wer​den; siehe „Anhang: Neue Befehle in Director 8“: on isOkToAttach).

Das Dialogfenster zeigt automatisch alle Bitmap-Darsteller in den Besetzungen zur Auswahl an:

Die folgende Tabelle zeigt alle verfügbaren Darstellertypen an. In der Spalte „the type of member“ steht, wel​chen Werten der Eigenschaft member("...").type (bzw. the type of member) der Datentyp ent​spricht.

	Format
	the type of member
	Beschreibung

	#bitmap
	#bitmap
	Bitmap-Darsteller

	#button
	#button
	Button-Darsteller

	#digitalVideo
	#digitalVideo (ohne #quickTimeMedia)
	Video-Darsteller, aber keine Quicktime-Darsteller

	#field
	#field (nicht #text)
	Field-Darsteller, aber keine Richtext-Darsteller

	#filmLoop
	#filmLoop
	Filmschleifen

	#graphic
	#animgif, #bitmap, #button, #digitalVideo, #flash, #field, #filmLoop, #movie, #ole, #picture, #text, #shape, #swa, #vectorShape
	Jeder Darsteller, der in einem Sprite-Kanal benutzt werden kann

	#member
	alle aus #graphic, plus #font, #palette, #script, #sound und #transition
	alle Darsteller

	#movie
	#movie
	Film-Darsteller

	#ole
	#ole
	OLE-Darsteller

	#palette
	#palette
	alle Director-Paletten plus Paletten-Darsteller der Besetzung

	#picture
	#picture
	PICT-Darsteller

	#richText
	#text
	Richtext-Darsteller

	#script
	#script
	Skripte

	#shape
	#shape
	Shape-Darsteller (Rechtecke, Ovale, Linien)

	#sound
	#sound (nicht #swa)
	Sound-Darsteller, aber kein Shockwave-Sound

	#transition
	#transition
	Alle Director-Transitions plus Transition-Darsteller der Besetzung


Unter Nicht-Darstellertypen fallen alle primitiven Lingo-Datentypen wie #integer oder #symbol, und all​gemeine Datentypen, z.B. #cursor.

	Format
	Werte
	Bemerkung

	#boolean
	TRUE und FALSE
	Wird als Checkbox angezeigt.

	#cursor
	verfügbare System-Cursor (keine Bitmap-Cursor)
	Cursor sind systemabhängig. Für Bitmap-Cursor #bitmap benutzen.

	#float
	Komma-Zahlen
	Es kann mit #min und #max ein Minimum und ein Maximum angegeben werden.

	#ink
	alle Inks
	Inks werden im Popup-Fenster als Text angegeben, intern aber als Ganzzahl entsprechend der Eigenschaft the ink of sprite gespeichert.

	#integer
	Ganzzahlen
	Es kann mit #min und #max ein Minimum und ein Maximum angegeben werden.

	#marker
	previous; loop; next und alle Marker im Drehbuch
	Previous, loop und next werden als Symbole gehandhabt, Drehbuchmarker als Zeichenketten.

	#string
	Zeichenkette
	

	#symbol
	Symbol
	


Alle Datentypen können in ihrer Beschreibung eine optionale #range-Eigenschaft definieren. Damit lässt sich der Auswahlbereich auf eine definierte Zahl von Einträgen eingrenzen. Die Werte werden in Form einer linea​ren Liste angegeben und lassen sich dann per Pull-Down-Menü auswählen.

Im folgenden Handler kann man beispielsweise nur eine der drei Sprachen „Deutsch“, „Englisch“ und „Franzö​sisch“ für die Eigenschaft mSprache auswählen:

on getPropertyDescriptionList
  description = [:]

  range = ["Englisch", "Französisch", "Deutsch"]

  description.addProp(#mSprache,¬

              [#comment: "Sprache:",¬

               #format: #string,¬

               #default: range[1],¬

               #range: range])

  return description

end getPropertyDescriptionList

Für die numerischen Datentypen Integer und Float lassen sich alternativ die minimal und maximal zuläs​sigen Werte angeben. Dazu wird #range eine weitere Eigenschaftsliste mit den Eigenschaften #min für den kleinsten und #max für den größten zulässigen Wert zugeordnet. Die Eingabe erfolgt dann per Schieberegler.

Als Beispiel soll hier noch mal unser Soundwiedergabe-Skript dienen. Hier sind für die Eigenschaft mSoundkanal nur ganzzahlige Werte zwischen 1 und 4 sinnvoll:

on getPropertyDescriptionList
  description = [:]

  description.addProp(#mTon, [ #comment: "Soundname ohne Endung:",  ¬

                               #format:  #string, ¬

                               #default: ""])

  description.addProp(#mSoundKanal, [ #comment: "SoundKanal:",  ¬

                                #format:  #integer, ¬

                                #default: 1,¬

                                #range:   [#min:1, #max:4] ])

  return description

end getPropertyDescriptionList

Eine weitere, jedoch undokumentierte Eigenschaft für Float-Werte ist #jump, die festlegt, um welchen Be​trag sich der Zeiger weiterbewegt, wenn auf den Schieberegler geklickt wird. Dieser Wert spiegelt sich in der Darstellung der Skala wider.

Man muss sich bei der Anzeige der Dialogfenster aber nicht auf Director verlassen. Es ist ebenso möglich, ein eigenes Dialogfenster zu öffnen, z.B. um Grafiken anzuzeigen. Dazu lässt sich das MUI-Xtra verwenden, das mit Director installiert wird.
 Der dazu notwenige Programmcode wird in den on runPropertyDialog-Handler geschrieben. Ist dieser vorhanden, öffnet Director nicht selbstständig ein Eingabefenster, sondern ruft diesen Handler auf. Als erstes Argument erhält er die Objektreferenz des Verhaltens (also immer me angeben!) , als zweites Argument die Liste, die der Handler on getPropertyDescriptionList zurückgibt. Als Rückgabewert sollte der Handler eine Eigenschaftsliste der Form [#Eigenschaftsname1, Eigenschaftswert1, ...] zurückliefern. Es ist natürlich auch möglich, überhaupt kein Fenster zu öffnen und z.B. feste oder errechnete Werte zurückzuliefern.

Bemerkung

Genaugenommen ist also nicht on getPropertyDescriptionList, sondern on runPropertyDialog für die Darstellung des Standard-Eingabedialoges zuständig. Intern passiert folgen​des: Wird ein Verhalten in das Drehbuch gezogen, ruft Director runPropertyDialog auf. Als Argument erhält dieser Hander eine Propertyliste, in der alle gesetzten Eigenschaften mit ihrem aktuellen Wert stehen. Daraufhin wird getPropertyDescriptionList aufgerufen um die Beschreibungsliste des Eingabe​fensters abzurufen. Die Standardimplementierung dieses Handlers erzeugt daraus mit Hilfe des MUI-Xtras das Eingabefenster. Wenn wir nun selbst runPropertyDialog definieren, überschreiben wird diese Methode. Die eigene Methode ersetzt folglich die von Director bereitgestellte.

5.7 Beispiel 9: Eigenschaften-Dialogfenster zur Auswahl einer Datei

Ein Vorteil der Auswahl von Werten aus einem Auswahlmenü im Eigenschaften-Dialogfenster ist, dass Fehler​eingaben fast ausgeschlossen sind. Wenn ein bestimmter Darsteller über ein Pulldown-Menü ausgewählt wird, ist sichergestellt, dass der entsprechende Darsteller auch wirklich existiert. Leider fehlt dem Director-eigenen Dialogfenster die Möglichkeit, eine externe Datei auszuwählen, beispielsweise wenn zu einem anderen Film gesprungen werden soll. Das folgende Skript realisiert eine solche Eingabe mit Hilfe des FileXtras, das stan​dardmäßig mit Director 7 installiert wird. Die Auswahl findet in einem gewöhnlichen Datei-Öffnen-Dialog statt. So ist sichergestellt, dass die ausgewählte Datei auch tatsächlich existiert. 

Wie bei einem Director-Eingabedialog müssen wir auch hier die benutzten Eigenschaften im getPropertyDescriptionList()-Handler definieren. Die Eigenschaft mMoviePath, die den Pfad des Films speichert, ist eine Zeichenkette, die auf den Anfangswert EMPTY gesetzt wird. Das eigentliche Öffnen des Datei-Fensters geschieht im runPropertyDialog()-Handler. Zuerst wird geprüft, ob der Film schon gespeichert wurde, da der ausgewählte Dateinamen als relativer Pfad zum aktuellen Film gespeichert werden soll. Auf diese Weise ist die Verknüpfung auch noch gültig, wenn das gesamte Projekt verschoben oder auf einem anderen Rechner installiert wird. Wenn noch kein Dateiname des aktuellen Films vorliegt, weist eine Warnmeldung den Benutzer darauf hin. Wurde der Film schon gespeichert, öffnet das Skript mit dem FileXtra Befehl FileOpenDialog ein Datei-Öffnen Fenster. Das angezeigte Verzeichnis wird auf das des aktuellen Fil​mes gesetzt, wozu die Methode pfadExtrahieren() benutzt wird. Der ausgewählte Dateiname wird dann mit der Methode generiereRelativenPfad() erzeugt, bei Klick auf den Abbrechen-Button wird der vorherige Wert über​nommen. Damit die Verknüpfung auf allen Plattformen funktioniert, wird der relative Pfad mit dem @-Opera​tor gebildet. Dadurch kann das Skript sowohl auf dem Mac wie auch unter Windows ohne Modifikation über​nommen werden.

Dies ist das vollständige Skript des go to movie...-Verhaltens

-- wechselt zu einem neuen Film; die Datei des neuen

-- Films wird mit Hilfe des Datei-Öffnen-Dialogs

-- des FileXtras ausgewählt; dabei wird automatisch ein

-- relativer Pfad generiert
property mMoviePath

on exitFrame me 

  go to movie mMoviePath

end exitFrame
on getPropertyDescriptionList
  set description = [:]

  description.addProp(#mMoviePath, [ #comment: "Neuer Movie:",  ¬

                                #format:  #string, ¬

                                #default: EMPTY])

  return description

end getPropertyDescriptionList
on runPropertyDialog me, propertyList

  if the moviePath = EMPTY then
    -- Film wurde noch nicht gespeichert

    alert("Der Film muss erst gespeichert werden!")

    return propertyList

  else
    if propertyList.mMoviePath = EMPTY then
      anzuzeigenderPfad = the moviePath
    else
      -- den Pfad aus mMoviePath extrahieren

      anzuzeigenderPfad = me.PfadExtrahieren(propertyList.mMoviePath)

    end if
    if the platform contains "Windows" then
      ergebnis = FileOpenDialog(anzuzeigenderPfad, "Director-Film/*.dir",¬

 "Neuen Film auswählen", FALSE, TRUE) 

    else

      -- auf dem Mac hier den benötigten Filetype eintragen 

      ergebnis = FileOpenDialog(anzuzeigenderPfad, "") 

    end if
    if (ergebnis = EMPTY) then
      -- Dialog wurde abgebrochen

      return propertyList

    else
      -- neuer Name wurde eingegeben

      neuerPfad = me.generiereRelativenPfad(ergebnis, me.getMoviePath())

      propertyList.mMoviePath = neuerPfad

      return propertyList

    end if
  end if
end runPropertyDialog
on generiereRelativenPfad me, pfad, relativZu

  -- gibt den Dateipfad pfad relativ zu Pfad relativZu 

  -- in unabhängigem Director-Format (@/.../...) zurück

  tempId = the itemdelimiter
  the itemdelimiter = me.trennzeichen()

  if pfad.item[1] = relativZu.item[1] then
    -- Pfad kann relativ angegeben werden

    repeat while pfad.item[1] = relativZu.item[1]

      delete pfad.item[1]

      delete relativZu.item[1]

    end repeat

    -- ermitteln, wieviele Ordner aufwärts gegangen werden muss (..)

    repeat with ebenen = 1 to (relativZu.items.count - 1)

      pfad = "/" & pfad 

    end repeat
    -- Trennzeichen durch "/" ersetzen

    repeat while pfad contains me.trennzeichen()

      put "/" into pfad.char[offset(me.trennzeichen(), pfad)]

    end repeat
    -- "@/" voranstellen

    pfad = "@/" & pfad

  else
    -- Pfad kann nicht relativ angegeben werden

    alert("Der Pfad kann nicht relativ angegeben werden!")

  end if
  the itemdelimiter = tempId

  return pfad

end generiereRelativenPfad

on pfadExtrahieren me, dateiname

  -- extrahiert aus dem angegebenen absoluten Dateinamen den Pfad

  tempId = the itemdelimiter
  the itemdelimiter = me.trennzeichen()

  delete the last item of dateiname

  the itemdelimiter = tempId

  return dateiname

end pfadExtrahieren

on trennzeichen me
  -- liefert das Plattformspezifische Pfadtrennzeichen zurück

  if the platform contains "Windows" then
    return "\"
  else
    return ":"
  end if
end trennzeichen

on getMoviePath me
  -- gibt den Moviepath zurück; es ist garantiert, dass der Pfad immer mit

  -- "\" bzw. ":" aufhört!

  if the last char of the moviePath = me.trennzeichen() then
    return the moviePath
  else
    return the moviePath & me.trennzeichen()

  end if
end getMoviePath

on getBehaviorDescription
  return "Springt zu einem neuen Movie."& RETURN ¬

  & "Der angegebene Pfad wird relativ zum aktuellen Film gespeichert." & RETURN ¬

  & "Achtung: Das FileXtra muss sich im Director-Ordner befinden!"
end getBehaviorDescription
5.8 Ermitteln des Sprites zu einem Verhalten mit the spriteNum
Viele Verhalten wirken auf das Sprite ein, dem sie zugeordnet sind. So verändert z.B. ein Rollover-Verhalten die Grafik des Sprites, auf dem es liegt. Das Wissen über die eigene Spritenummer ist für eine Reihe von Funktionen wichtig, z.B. auch Drag-And-Drop-Interfaces, Buttons u.s.w. Aus diese Grunde bietet jedes Ver​halten standardmäßig die Eigenschaft spriteNum, die die Nummer des Sprites eines Verhaltens speichert. Im folgenden ist der Anfang eines Rollover-Verhaltens zu sehen, das am Anfang des Sprites die aktuelle Grafik in der Eigenschaft mMemberNormal speichert, um diese anzuzeigen, nachdem die Maus das Sprite wieder verlas​sen hat.

property mMemberNormal, mMemberRollover

on beginSprite me
 mMemberNormal = sprite(me.spritenum).member
end beginSprite
Die Konstruktion mit me (me.spriteNum  bzw. the spritenum of me) ist nicht nötig, wenn spriteNum als Eigenschaft dieses Verhaltens definiert wird.

property mMemberNormal, mMemberRollover, spriteNum
on beginSprite me
 mMemberNormal = sprite(spriteNum).member
end beginSprite
Zusätzlich zu spriteNum gibt es noch die Film-Eigenschaft the currentSpriteNum. Sie zeigt an, wel​ches Skript gerade abgearbeitet wird. Es ist in Darsteller- und Sprite-Skripten gültig; in Frame- und Film​skripten liefert es 0 zurück. Diese Eigenschaft war vor allem vor der Einführung von Verhalten sinnvoll, um verhaltensähnliche
 Skripte zu schreiben. Es wird in Verhalten nur noch selten gebraucht, z.B. in on getPropertyDescriptionList, weil spriteNum bei Aufruf dieses Handlers noch nicht definiert ist.

5.9 Senden von Nachrichten mit sendSprite und sendAllSprites
Im Gegensatz zu Parent-Skripten wird die Objektreferenz auf ein Verhalten in der Regel nicht in einer Variab​len gespeichert, sondern ist einem bestimmten Sprite zugeordnet. Mit dem Befehl sendSprite bietet Lingo die Möglichkeit, gezielt die Methoden der Verhalten eines Sprites aufzurufen. Die Syntax ähnelt dem call-Befehl:

sendSprite (Spritenummer, #handlerName, {argumente...})

Spritenummer bezeichnet die Nummer des Sprites, dessen Methode aufgerufen werden soll. Der Name des Handlers wird, genau wie bei call und callAncestor, als Symbol angegeben. Optional können der Me​thode Argumente übergeben werden. Achtung: Auch beim Methodenaufruf mit sendSprite wird als ersters Argument die Objektreferenz (me) übergeben.

Der Aufruf 

sendSprite (1, #ausgabe, "Ich bin’s.")

ruft die Methode im Verhalten von Sprite 1 auf:

on ausgabe me, ausgabeText

  put "Sprite" && me.spritenum && "meldet:" && ausgabeText

end ausgabe

Die Ausgabe im Nachrichtenfenster lautet:

-- Sprite 1 meldet: Ich bin’s.

Der Methodenaufruf mit sendSprite hält sich an die übliche Reihenfolge bei der Abarbeitung von Ereignis​sen (Events). In der Regel wird die auszuführende Methode in allen Verhalten ausgeführt, die einem Sprite zugeordnet sind. Ist keine entsprechende Methode vorhanden, wird im Darstellerskript, Frameskript und schließlich im Filmskript weitergesucht, wobei ein vorhandener Handler die Eventkette beendet. Mit dem Be​fehl pass lässt sich der Aufruf explizit an die jeweils nächste Ebene weitergeben. Entsprechend kann mit dem Befehl stopEvent verhindert werden, dass der Aufruf von einem Verhalten an ein anderes weitergegeben wird (siehe „Das Director Event-Modell“).

Wenn keine Methode bzw. kein Handler mit dem angegebenen Namen existiert, wird keine Fehlermeldung ausgegeben.

Obwohl der Name des sendSprite-Befehls nahe legt, dass es sich um eine einseitige Kommunikation han​delt, können die Methoden durchaus Werte zurückliefern. Dies ist z.B. beim Abfragen von Eigenschaften sinn​voll. Die folgende Methode liefert den ancestor des Verhaltens zurück.

on getAncestor me
  return ancestor
end getAncestor

Folgende Zeile speichert den ancestor dieses Verhaltens auf Sprite 3 in der Variablen thisAncestor:
thisAncestor = sendSprite(3, #getAncestor)

Sind mehrere Verhalten mit dem gleichen Methodennamen einem Sprite zugeordnet, wird der Wert des zuletzt aufgerufenen Verhaltens zurückgeliefert. Ist keine Methode mit entsprechendem Namen vorhanden, wird VOID zurückgeliefert.

Die Eigenschaften eines Verhaltens lassen sich auch über

sprite(Spritenummer).Eigenschaftsname
abfragen. In diesem Fall muss diese Eigenschaft aber auf jeden Fall vorhanden sein, da Lingo ansonsten einen Fehler meldet. Im Gegensatz zum Ermitteln eines Methodenwertes mit sendSprite wird nicht der letzte, sondern der erste verfügbare Wert zurückgeliefert.

Analog dazu ist es auch möglich, eine Methode über die Syntax

sprite(Spritenummer).Methodenname()

aufzurufen. Dies entspricht einem Aufruf durch sendSprite mit folgenden Unterschieden:

1. Die Methoden müssen vorhanden sein, sonst meldet Director einen Fehler.

2. Der Methodenaufruf wird nicht in der Ereigniskette weitergereicht: es werden also nur die Methoden des der Sprite-Verhalten aufgerufen, nicht des Darsteller- oder Filmskriptes.

Bemerkung

Director erzeugt Verhaltensinstanzen erst im laufenden Film. Daher können Eigenschaftwerte von Verhalten in einem angehaltenen Film weder direkt, noch über eine Methode abgefragt werden, da sie noch nicht initialisiert sind. Die Werte, mit denen die Eigenschaften initialisiert werden, sind im Verhaltensinspektor ersichtlich. Sollte es nötig sein, diese Werte per Lingo zu ermitteln, kann dazu der undokumentierte Befehl the scriptList of sprite benutzt werden.

Mit dem Befehl sendAllSprites wird eine Methode in allen Sprites eines Frames aufgerufen. Er arbeitet ansonsten genauso wie sendSprite.

sendAllSprites (#handlerName, {argumente...})

Bemerkung

Methodenaufrufe mit sendSprite/sendAllSprites und das direkte Abfragen von Verhaltenseigen​schaften können sich auf mehrere Objekte beziehen und sind daher nicht eindeutig. Dies kann zu Fehlern füh​ren, die schwer zu finden sind. Außerdem ist es schwieriger als bei Child-Objekten, Eigenschaftswerte abzufra​gen und zu überwachen. Daher ist es ratsam, Verhalten vor allem für das Verarbeiten von Ereignissen einzu​setzen. Ein Vorschlag für eine sinnvolle Kombination von Verhalten und Objekten findet sich im Unterkapitel „Kommunikation zwischen Verhalten und Objekten“.

5.10 Ermitteln der Verhalten zu einem Sprite mit the scriptInstanceList of sprite und the scriptList of sprite
Mit der Eigenschaft the scriptInstanceList of sprite erhält man im einem laufenden Film eine Liste der Verhaltensinstanzen eines bestimmten Sprites. Diese hat die Form einer linearen Liste und enthält die Verhaltensreferenzen.

put sprite(14).scriptInstanceList

-- [<offspring "hot/nothot & cursor-skript" 1 112194c>, <offspring "mouseup an Szene-Objekt" 1 1121960>]

Da die Liste nur Instanzen enthält, ist sie leer, wenn ein Film gerade nicht läuft. Der Liste können auch neue Instanzen hinzugefügt werden, sie muss dann aber schon mindestens einen Eintrag enthalten. Einem Sprite, dem noch kein Verhalten zugeordnet wurde, kann also kein Verhalten hinzugefügt werden. Zu beachten ist außerdem, dass die Änderungen der scriptInstanceList nicht von bleibender Dauer sind, sondern nur bis zum Ende des Sprites gelten.

Diese Eigenschaft lässt sich beispielsweise dazu verwenden, eine Methode eines bestimmten Verhaltens aufzu​rufen ohne gleichlautende Methoden des selben Sprites versehentlich mit anzusprechen. Folgende Zeilen rufen die Methode setDebugMode() des zweiten Verhaltens von Sprite 20 auf, ohne dass eine gleichlautende Methode im ersten Verhalten zu einem Konflikt führt:

thisInstance = sprite(20).scriptInstanceList[2]

call(#setDebugMode, thisInstance, TRUE)

Im Gegensatz zu the scriptInstanceList enthält the scriptList of sprite nicht die Skriptinstanzen, sondern die Skriptdarsteller und Initialisierungseigenschaften eines Sprites, also die Informa​tionen, die im Verhaltensinspektor angezeigt werden. 

put sprite(14).scriptList

-- [[(member 15 of castLib 3), "[#mSpriteCursor: "Finger"]"], [(member 13 of castLib 3), "[#mSymbol: #leonore]"]]

In Director 7 lässt sich die Skriptliste nicht bearbeiten, sie dient nur zu Informationszwecken.

Bemerkung

In Director 7 ist the scriptList of sprite eine undokumentierte Eigenschaft; sie hat mit Version 8 Einzug in den Lingo-Wortschatz gehalten. Zusätzlich ist es möglich geworden, dieser Liste mit dem Befehl setScriptList eigene Werte zuzuweisen (siehe „Anhang: Neue Befehle in Director 8“).

5.11 Kommunikation zwischen Verhalten und Objekten

Die Möglichkeit, in Lingo objektorientiert zu programmieren gibt es in der jetzigen Form seit Director 5. Al​lerdings kommen Objektskripte in der Praxis nicht oft zum Einsatz: Director wird zum Großteil von Multi​media-Autoren, nicht von gelernten Informatikern programmiert, und die schätzen an Director vor allem die visuelle Arbeitsweise. In Programmiersprachen ohne ein visuelles Interface, z.B. Java oder C++, wird dort, wo in Lingo das Drehbuch das Hauptwerkzeug ist, auf Skripting gesetzt, d.h. Animationen werden in einer exter​nen Datei oder internen Ressource definiert, häufig in von Menschen lesbarem Textformat. Diese Möglichkeit bietet Director natürlich auch; aus einer Text-Datei bzw. einem Textdarsteller können z.B. die Eigenschaften bestimmter Variablen oder Objekte generiert werden. Das Einlesen dieser Dateien wird auch als Parsing be​zeichnet. Diesen Parser, also der Programmteil, der das Auslesen übernimmt, kann man entweder selbst schreiben oder den seit Director 7 mitgelieferten XML-Parser benutzen. XML ist ein flexible Datenbeschrei​bungsformat, das HTML, also dem Datenbeschreibungsformat des World Wide Web ähnelt. Auf XML näher einzugehen, würde den Rahmen dieser Arbeit sprengen. Wenn man jedoch auf das schnelle Auslesen größerer, mehr oder weniger abstrakter Datenmengen angewiesen ist, ist XML für Director die richtige Wahl.

XML kann so aussehen, wie dieser Ausschnitt einer XML Datei aus einem Phantombildgenerator-Programm. Es definiert die verfügbaren Einzelteile, ihren Zusammenhang, ihre Tiefeninformation etc.

<?xml version="1.0"?>

<PHANTOMBILDER>

<SET NAME="Mensch">

<!-- Augen -->

<ELEMENT>

<MEMBER>A-augen1</MEMBER>

<Z-INDEX>80</Z-INDEX>

<GRUPPE>Augen</GRUPPE>

<DEFAULTLOC X="-50" Y="27"/>

<BRUDER>

<MEMBER>B-augen1</MEMBER>

<Z-INDEX>80</Z-INDEX>

<DEFAULTLOC X="51" Y="26"/>

</BRUDER>

<EXKLUSIV/>

<AUTOSCALE/>

<AUTOROTATE/>

</ELEMENT>

<ELEMENT>

<MEMBER>A-augen2</MEMBER>

<Z-INDEX>80</Z-INDEX>

<GRUPPE>Augen</GRUPPE>

<DEFAULTLOC X="-45" Y="19"/>

<BRUDER>

<MEMBER>B-augen2</MEMBER>

<Z-INDEX>80</Z-INDEX>

<DEFAULTLOC X="46" Y="18"/>

</BRUDER>

<EXKLUSIV/>

<AUTOSCALE/>

<AUTOROTATE/>

</ELEMENT>

...

Wenn immer möglich ist in Director die Montage von Layouts und Animationen im Drehbuch dem Skripting vorzuziehen. Die Schnittstelle zwischen Drehbuch und Skript stellen die Verhalten dar: sie kommunizieren Interaktion an das Programm weiter. Verhalten sind Objekte, die genau wie Parent-Skripte eingesetzt werden können. Ihr entscheidender Nachteil ist jedoch ihre Anonymität: Sie werden vom Drehbuch generiert und las​sen sich nur umständlich direkt ansprechen. Child-Objekte sind dagegen direkt über eine Referenzvariable mit aussagekräftigem Namen zugänglich. Sie eignen sich generell besser für die Sammlung von Daten und Funk​tionen. Verhalten dagegen sind Spezialisten in der Verarbeitung von Ereignissen. Es liegt daher nahe, beide zu koppeln: die Ereignisse werden von den Verhalten verarbeitet und diese wiederum erzeugen Objekte bzw. geben Werte an Objekte weiter. Zum Beispiel könnte ein Verhalten auf einer Spielplangrafik am Programm​anfang die Spritenummer und die Dimensionen des Spielplans an ein Spielplanobjekt kommunizieren; diese Werte müssten nicht hart kodiert werden und bei einem Wechsel der Größe des Spielfelds oder der Verschie​bung um einige Spritekanäle müsste keine Zeile Programmcode geändert werden.

Das folgende Beispiel soll die Vorteile dieser Vorgehensweise verdeutlichen. 

Das Parent-Skript BallObjekt bewegt ein Sprite auf der Bühne auf einer geraden Bahn; wenn es am Rand eines Rechtecks angekommen ist, prallt es an der jeweiligen Kante ab. Dieses Rechteck ist mit festen Werten pro​grammiert; der Abprall-Algorithmus ist nicht sonderlich ausgefeilt, es wird lediglich die Bewegungsrichtung umgekehrt, doch für dieses Beispiel ist dies ausreichend. Als Parameter sind das Sprite zu übergeben und die anfängliche Richtung als Bewegungsvektor in Punktform (point(dx, dy)). Da die Bewegung in einem stepframe-Handler stattfindet, ist das Objekt der actorlist hinzuzufügen.

property mSprite, mBewegungsvektor

on new me, spriteNummer, anfangsVektor

  -- Sprite-Kanal

  mSprite = spriteNummer

  -- anfänglicher Bewegungsvektor: Werte aus Behavior einlesen

  mBewegungsvektor = anfangsVektor

  return me
end new
on stepframe me
  newLoc = me.calculateNewLoc()

  if (newLoc.locH < 75) or (newLoc.locH > 515) then
    -- Ball prallt von senkrechter Linie ab

    mBewegungsvektor = mBewegungsvektor * point(-1, 1)

    newLoc = me.calculateNewLoc()

  end if
  if (newLoc.locV < 75) or (newLoc.locV > 355) then
    -- Ball prallt von waagerechter Linie ab

    mBewegungsvektor = mBewegungsvektor * point(1, -1)

    newLoc = me.calculateNewLoc()

  end if
  sprite(mSprite).loc = newLoc

end stepframe
on calculateNewLoc me
  return (sprite(mSprite).loc + mBewegungsvektor)

end calculateNewLoc

Das Erzeugen der Objekte kann in einem startMovie-Handler stattfinden:

on startMovie
  clearGlobals
  (the actorList).add(script("BallObjekt", 1, 30, 40)

  (the actorList).add(script("BallObjekt", 2, 10, -10)

  (the actorList).add(script("BallObjekt", 3, -1, -4)

end startMovie
Hier werden 3 Objekte erzeugt, die Sprite 1 bis 3 bewegen. Der clearGlobals-Befehl hat die Funktion, die actorList von eventuell aus einem vorherigen Programmablauf erzeugten Objekten zu befreien.

Mit dieser Vorgehensweise haben wird jedoch das Problem, dass bei jeder Verschiebung eines Sprites das Skript aktualisiert werden muss und ebenso, wenn ein neues Sprite hinzugefügt wird.

Das BallVerhalten schafft hier Abhilfe: es erzeugt selbständig ein neues Objekt mit dem richtigen Sprite und entfernt es, sobald das Sprite endet (das schließt das Beenden des Films mit ein). Das Hinzufügen eines neuen Balles ist denkbar einfach: das Verhalten wird einfach auf den Darsteller auf der Bühne gezogen. Als Drein​gabe kann der Anfangsbewegungsvektor komfortabel in einer Dialogbox mithilfe von Schiebereglern eingestellt werden.

property spriteNum, mObjekt, mSpeedX, mSpeedY

on beginSprite me
  mObjekt = script("BallObjekt").new (spriteNum, point(mSpeedX, mSpeedY))

  (the actorlist).add(mObjekt)

end beginSprite
on endSprite me
  (the actorlist).deleteOne(mObjekt)

end endSprite
on getPropertyDescriptionList me
  description = [:]

  addProp     description,#mSpeedX,[#default:0,¬

#format:#integer,#comment:"horizontale Geschwindigkeit:",¬

 #range:[#min:-20, #max: 20]]

  addProp     description,#mSpeedY,[#default:0,¬

#format:#integer,#comment:"vertikale Geschwindigkeit:",¬

 #range:[#min:-20, #max: 20]]

  return description

end getPropertyDescriptionList
on getBehaviorDescription me
  return "Behaviour für den springenden Ball."
end getBehaviorDescription
Natürlich wäre es auch möglich, die gesamte Funktion in einem einzigen Verhalten zu vereinen. Das würde das Programm jedoch unnötig verkomplizieren. Objektorientierte Programmierung bietet die Möglichkeit der Abstraktion, und das Aufteilen in eine Ereignisebene und eine Funktionsebene stellt ebene eine solche dar.

5.12 Beispiel 10: Erweiterung des Ball-Beispiels

Das abschließende Beispiel zeigt, wie man bei unserem Ball-Beispiel aus dem letzten Unterkapitel auch die Dimensionen des Rechtecks dy​namisch über ein Verhalten setzen kann. Es ist ohne weiteres möglich, das Rechteck im laufenden Programm zu verschieben, indem man es auf moveable setzt.

Das Rechteck-Objekt definiert dieses Rechteck, dessen einzige Aufgabe es ist, die Koordinaten der Rechteck-Ränder zu ermitteln (getLeft(), getRight(), getTop(), getBottom()). Über die Methode spriteAnmelden() teilt das Rechteck-Verhalten mit, in welchem Spritekanal das Rechteck liegt. Die Methode isReady() stellt fest, ob schon eine Anmeldung erfolgt ist.

-- das Rechteck, das als Begrenzer für die springenden Bälle dient

-- es muss ein Sprite angemeldet werden, damit das Skript funktioniert

property mSprite

on new me
  -- Rechtecksprite ist anfangs noch undefiniert

  mSprite = VOID 

  return me
end new
on spriteAnmelden me, spriteNummer

  -- meldet das Rechteck-Sprite an

  mSprite = sprite(spriteNummer)

end spriteAnmelden

on getLeft me
  -- gibt die x-Koordinate der linken Kante des Rechtecks zurück

  if me.isReady() then
    return mSprite.locH
  else
    return VOID
  end if
end getLeft

on getRight me
  -- gibt die x-Koordinate der rechten Kante des Rechtecks zurück

  if me.isReady() then
    return mSprite.locH + mSprite.width
  else
    return VOID
  end if
end getRight

on getTop me
  -- gibt die y-Koordinate der oberen Kante des Rechtecks zurück

  if me.isReady() then
    return mSprite.locV
  else
    return VOID
  end if
end getTop

on getBottom me
  -- gibt die y-Koordinate der unteren Kante des Rechtecks zurück

  if me.isReady() then
    return mSprite.locV + mSprite.height
  else
    return VOID
  end if
end getBottom

on isReady me
  -- prüft, ob das Rechteckobjekt schon initialisiert wurde,

  -- d.h. mSprite gesetzt ist

  return objectP(mSprite)

end isReady

Das zugehörige Rechteck-Verhalten setzt voraus, dass das Rechteck-Objekt schon initialisiert und der Globalen gRechteckObj zugewiesen wurde, was in unserem Beispiel im prepareMovie-Handler geschieht.

-- Verhalten, dass das begrenzende Rechteck anmeldet;

-- auf ein Sprite ziehen, das als Begrenzer für

-- die Bälle dienen soll
global gRechteckObj

property spriteNum
on beginSprite me
  -- dieses Sprite beim Rechteck-Objekt anmelden

  gRechteckObj.spriteAnmelden(spriteNum)

end beginSprite
on getBehaviorDescription me
  return "Dieses Verhalten auf ein Sprite ziehen, das" & ¬

"als Begrenzer für die hüpfenden Bälle dienen soll."
end getBehaviorDescription

Das Skript zum neuen BallObjekt sieht nun wie folgt aus:

global gRechteckObj

property mSprite, mBewegungsvektor

on new me, spriteNummer, anfangsVektor

  -- Sprite-Kanal

  mSprite = sprite(spriteNummer)

  -- anfänglicher Bewegungsvektor: Werte aus Behavior einlesen

  mBewegungsvektor = anfangsVektor

  return me
end new
on stepframe me
  newLoc = mSprite.loc + mBewegungsvektor

  -- aktuelle Kanten des Balls bestimmen

  linkeKante = newLoc[1] - mSprite.member.regpoint[1]

  rechteKante = newLoc[1] + mSprite.member.width - mSprite.member.regpoint[1]

  obereKante = newLoc[2] - mSprite.member.regpoint[2]

  UntereKante = newLoc[2] + mSprite.member.height - mSprite.member.regpoint[2]

  -- Kollission kann nur geprüft werden, wenn Rechteckobjekt

  -- schon initialisiert ist

  if gRechteckObj.isReady() then
    if (linkeKante < gRechteckObj.getLeft()) then
      -- Ball prallt von linker Kante ab

      -- x-Koordinate an linker Kante spiegeln

      newLoc[1] = newLoc[1] - 2*(linkeKante - gRechteckObj.getLeft())

      -- Bewegungsrichtung umkehren

      mBewegungsvektor = mBewegungsvektor * point(-1, 1)

    else
      if (rechteKante > gRechteckObj.getRight()) then
        -- Ball prallt von rechter Kante ab

        -- x-Koordinate an rechter Kante spiegeln

        newLoc[1] = newLoc[1] - 2*(rechteKante - gRechteckObj.getRight())

        -- Bewegungsrichtung umkehren

        mBewegungsvektor = mBewegungsvektor * point(-1, 1)

      end if
    end if
    if (obereKante < gRechteckObj.getTop()) then
      -- Ball prallt von oberer Kante ab

      -- y-Koordinate an oberer Kante spiegeln

      newLoc[2] = newLoc[2] - 2*(obereKante - gRechteckObj.getTop())

      -- Bewegungsrichtung umkehren

      mBewegungsvektor = mBewegungsvektor * point(1, -1)

    else
      if (untereKante > gRechteckObj.getBottom()) then
        -- Ball prallt von unterer Kante ab

        -- y-Koordinate an unterer Kante spiegeln

              newLoc[2] = newLoc[2] - 2*(untereKante - gRechteckObj.getBottom())

        -- Bewegungsrichtung umkehren

        mBewegungsvektor = mBewegungsvektor * point(1, -1)

      end if
    end if
  end if
  -- Sprite aktualisieren

  mSprite.loc = newLoc

end stepframe
Ein kleiner Trick wurde bei dem Darsteller angewendet, der den erklärenden Text enthält. Damit die schwarze Schrift immer innerhalb des Rechtecks liegt und lesbar bleibt, wurde einfach ein Ball-Verhalten mit dem Be​wegungsvektor (0|0) am Textdarsteller angebracht.

Beispiele

Dieser Teil der Arbeit beschreibt die Funktionsweise von drei vollwertigen Spielen, die von der objektorien​tierten Programmierung Gebrauch machen. Damit die Strukturen nachvollziehbar bleiben, sind die Beispiele bewusst einfach gehalten und konzentrieren sich auf das Wesentliche. Professionelle Director-Spiele haben in der Regel den folgenden Umfang an Lingo-Skripten:

Internet-Spiel







500 – 1500 Zeilen

Einfaches nonlineares Grafikadventure




ca. 5000 Zeilen

Spiel mit weitgehenden Interaktionsmöglichkeiten (z.B. Zeichenfunktionen)
> 10 000 Zeilen

In den Beispielen will ich vor allem auf die Strukturierung und Planung der Programmierung eingehen, nicht auf die Programmierung selbst. Details zur Programmierung entnimmt man am besten dem kommentierten Quellcode selbst.

5.13 3×3-Puzzle

5.13.1 Beschreibung

Aus 9 Puzzleteilen soll ein 3x3 großes Quadrat gelegt werden. Auf jedem Puzzleteil sind 4 Symbole zu sehen, die an den Kanten horizontal oder vertikal zerschnitten wurden. Ziel des Spieles ist, die Teile so zusammenzu​setzen, dass wieder vollständige Figuren entstehen.

5.13.2 Definition der Objekte

Ein Spiel besteht immer aus 9 Puzzleteilen. Es sorgt dafür, dass alle Puzzleteile gezeichnet werden, indem es den Aufruf von draw() an alle Puzzleteile weiter delegiert. Dabei ordnet es jedem Puzzleteile einen Spritekanal zu. Durch die Methode obenEinsortieren() kann die Tiefensortierung beeinflusst werden. Jedes Puzzleteil, das gezogen oder gedreht wird, wird an die oberste Stelle einsortiert. Die entstehende Lücke wird durch die nach​rückenden Puzzleteile wieder aufgefüllt. Die Reihenfolge der Puzzleteile wird durch ihre Position in der Liste mPuzzleteileOrdnung bestimmt, die bei jedem Zeichnen neu ausgewertet wird.

Ein Puzzleteil ist das auf der Spielfläche sichtbare „Kärtchen“. Es zeichnet sich durch seine Position und seine aktuelle Drehung aus. Entsprechend definiert es Methoden, um diese Eigenschaften zu manipulieren, nämlich die Methode moveTo(), um das Teil an eine neue Position zu schieben, sowie die Methode rotate(), mit der es um 90 Grad im Uhrzeigersinn gedreht wird. Da es per Klicken und Ziehen mit der Maus auf dem Bildschirm bewegt wird, definiert es außerdem die vom „Drag & Drop“-Objekt benötigten Methoden. Das Puzzleteil reprä​sentiert die sichtbare Darstellung eines Spielkärtchens auf dem Bildschirm und implementiert folglich die Methode draw(). Es speichert die anzuzeigenden Grafiken allerdings nicht selbst, sondern erbt seine Beschrei​bung von einem Element-Objekt. Den entsprechenden Grafik-Member für ein Symbol erhält es vom Grafik​manager-Objekt.

Ein Element stellt die logische Einheit des Puzzles dar. Es speichert, in welcher Reihenfolge die Symbole auf den Kärtchen vorkommen und ob diese horizontal oder vertikal durchgeschnitten sind. Es gelten folgende Re​geln:

· Es gibt in einem Spiel immer 4 verschiedene Symbole.

· Diese können horizontal oder vertikal durchgeschnitten sein.

· Jedes Element enthält alle 4 Symbole, es darf keines doppelt vorkommen.

Folglich gibt es für die Symbole nur 6 verschiedene Reihenfolgen:

[image: image14.emf]
Jedes Symbol kann in einer von vier Ausrichtungen vorkommen:

[image: image15.emf]
Daraus ergeben sich 6 × 4 × 4 × 4 × 4 = 1536 verschiedene Elemente. Ein Element ist durch seine Ordnungs​nummer also hinreichend beschrieben. Deshalb hat der new-Handler des Element-Objekt als Argument nur diese Nummer. Aus ihr kann dann bestimmt werden, wie das Element aufgebaut ist. Dabei wird es intern so gehandhabt, dass die Elemente 0-255 die Reihenfolge ABCD in allen möglichen Ausrichtungskombinationen enthalten, Element 256 – 511 die Elemente der Reihenfolge ABDC, u.s.w. Daraufhin werden die 4 Symbol-Objekte konstruiert und gespeichert.

Aufgrund dieser Systematisierung lässt sich jedes Spiel als Liste von 9 Ganzzahlen speichern und in einer Skriptdatei ablegen. Das erste Spiel der Skriptdatei sieht beispielsweise so aus:

[19,235,292,731,404,488,1429,1112,407]

Ein Symbol-Objekt speichert, welches der 4 möglichen Symbole dargestellt wird, und ob es sich um den obe​ren, unteren, rechten oder linken Teil der Grafik handelt. Welche Grafikdarsteller dies genau sind, ermittelt das Grafikmanager-Objekt. Jede Grafik wird dazu in 4 Teile zerschnitten (oben links, oben rechts, unten links und unten rechts). Der Grafikmanager ermittelt zu einem gegebenen Symbol genau die Grafiken, die zusam​mengesetzt das richtige Bild ergeben. Als Argument des new-Handlers erhält es dabei eine Liste der zu ver​wendenden Grafiken aus einer Skriptdatei. Die Symbolgrafik kann im Spiel schnell geändert werden, indem ein Grafikmanager mit einer anderen Liste von Grafiken initialisiert wird. Eine Liste kann zum Beispiel so aussehen:

[[#lo:"sea 1 lo",#ro:"sea 1 ro",#lu:"sea 1 lu",#ru:"sea 1 ru"], 
[#lo:"sea 2 lo",#ro:"sea 2 ro",#lu:"sea 2 lu",#ru:"sea 2 ru"], 
[#lo:"sea 3 lo",#ro:"sea 3 ro",#lu:"sea 3 lu",#ru:"sea 3 ru"], 
[#lo:"sea 4 lo",#ro:"sea 4 ro",#lu:"sea 4 lu",#ru:"sea 4 ru"]]

Das Symbol-Objekt kann mit der Methode istGegenstueck() ermitteln, ob es mit einer gegebenen Symbolhälfte ein vollständiges Symbol ergibt. Ist dies für alle Symbole der Fall, ist das Puzzle gelöst.

Objektverhältnisse des 3×3-Puzzles

5.13.3 Besonderheiten

Die Puzzleteile werden vom Benutzer mit der Maus per „Drag & Drop“ bewegt. Wird ein Teil nur angeklickt und nicht gezogen, dreht es sich um 90° im Uhrzeigersinn. Ebenso kann die 3×3 Felder große Montagefläche an einer noch nicht belegten Stelle angefasst werden und alle schon zusammengesetzten Puzzleteile auf einmal verschoben werden. Aus diesem Grund habe ich das allgemein gehaltenes „Drag & Drop“-Objekt DragObjekt geschrieben, das für beide Zwecke eingesetzt werden kann. 

Der new-Handler des DragObjektes wird mit zwei Argumenten aufgerufen: das erste Argument ist das Objekt, das gezogen werden soll, das zweite ist die Zahl der Pixel, ab denen das Objekt beim Ziehen nicht mehr als angeklickt gelten soll. Durch eine höhere Zahl wird die Klickabfrage toleranter und ein leichtes Abrutschen des Cursors beim Anklicken wird verziehen. Das Ziehen des Objekts wird mit dem Befehl start() eingeleitet. Mit stop() wird das Ziehen beendet und es erfolgt die Auswertung der Aktion. Ein einfacher Aufruf kann z.B. so aussehen:

property mDragObjekt

on mouseDown me
  mDragObjekt = script("DragObjekt").new(me, 5) -- 5 Pixel Toleranz
  mDragObjekt.start()  -- ab jetzt werden die Mausereignisse ausgewertet
end mouseDown
on mouseUp me
  if objectP(mDragObjekt) then
    mDragObjekt.stop()  -- beenden des Ziehens
    mDragObjekt = 0 -- Objekt wird nicht mehr benötigt
  end if
end mouseUp
Das DragObjekt erwartet vom gezogenen Objekt, dass es folgende Methoden implementiert (Callback-Prinzip):

	on click me
	Diese Methode wird nach dem Aufruf der Methode stop() ausgeführt, wenn das Objekt angeklickt wurde, also nicht über dem Toleranzbereich hinausbewegt wurde.



	on drop me
	Diese Methode wird nach dem Aufruf der Methode stop() ausgeführt, wenn das Objekt über den Toleranzbereich hinausgezogen wurde.



	on startDrag me
	Diese Methode wird einmal aufgerufen, sobald das Objekt über den Toleranzbereich hinaus gezogen wurde.



	on drag me, x, y
	Diese Methode wird in jedem Frame aufgerufen, in dem das Objekt gezogen wurde. x und y sind die Koordinaten, an die das Objekt gezogen wurde.



	on getLoc me
	Diese Methode muss die aktuelle Position des Objektes zurückgeben.




Das letzte Objekt im Puzzle-Programm ist das SpielfeldObjekt. Es speichert die Größe des Spielfelds und wel​che Puzzleteile an welchen Positionen liegen. Außerdem prüft es, ob das Puzzle gelöst wurde, sobald alle Puzzleteile auf dem Spielfeld liegen. Die Dimensionen des Spielfeldes erfährt das SpielfeldObjekt vom Spiel​feld-Verhalten, das auf einem unsichtbaren Darsteller über dem Spielfeld auf der Bühne liegt, welcher auch den aktiven Bereich für Mausklicks über dem Spielfeld darstellt.

Screenshot des 3×3-Puzzles
5.14 Sokoban

5.14.1 Beschreibung

Sokoban (japanisch für „Lagerarbeiter“) ist ein klassisches Computerspiel aus den 1980er Jahren. Ziel des Spieles ist es, in einem Labyrinth verstreute Kisten auf vorgegebene Zielfelder zu schieben. Erschwert wird dies dadurch, das der Sokoban die Kiste nur schieben, nicht ziehen kann. Außerdem kann immer nur eine Kiste auf einmal bewegt werden. Im Gegensatz zum Originalspiel erfolgt die Spielsteuerung nicht mit der Tastatur, son​dern mit der Maus. Wenn ein direkter Weg möglich ist, reicht es, auf eine freies Feld im Labyrinth zu klicken und die Spielfigur findet automatisch den kürzesten Weg. Ob ein Feld erreichbar ist oder eine Kiste verschoben werden kann, wir durch eine Veränderung des Mauszeigers dargestellt.

Screenshot des Sokoban-Spiels (Level 8)

5.14.2 Definition der Objekte

Das Objekt, das für den gesamten Spielablauf zuständig ist, ist der Spielmanager. Seine Hauptaufgabe ist das Erzeugen eines neuen Spieles, wobei es das Sokoban-Objekt und den Kistenverwalter initialisiert. Es realisiert auch den Neustart eines Spieles sowie den Wechsel zu einem neuen Level. Dabei speichert es den Zustand des aktuellen Levels, damit bei der Rückkehr dieser Spielstand wieder geladen werden kann Außerdem delegiert es den Undo-Befehl und den Befehl, die aktuelle Position in einer History aufzuzeichnen, an die anderen Objekte.

Die Beschreibung der einzelnen Levels liest das LevelParser-Objekt aus einem Textdarsteller. Die Form dieser Beschreibung lehnt sich an die üblichen Konventionen zur Beschreibung eines Sokoban-Levels; es können also auch Levels aus dem Internet geladen werden und diesem Darsteller hinzugefügt werden. Level 2 sieht bei​spielsweise folgendermaßen aus:

"L e v e l   2"

#######

#  $ .#

#  @$ #

#  #$.#

#  . ##

######

Die genaue Beschreibung, wie ein Level definiert wird, befindet sich im Textdarsteller „Levels“. Das Level​Parser-Objekt wertet diesen Darsteller aus und ermittelt die Anzahl der Levels, ihren Aufbau und den Namen eines Levels.

Das Spielfeld-Objekt generiert aus der Levelbeschreibung in Textform ein Raster in Form einer verschachtelten Liste, an dessen Positionen es speichert, ob ein Feld leer ist, ein Zielfeld darstellt oder durch eine Wand ver​stellt wird. Aus dieser Information wird das Spielfeld generiert als Grafik in einer Variablen gespeichert. Das Spielfeldobjekt vereinigt alle Funktionen, die unmittelbar mit dem Aufbau und der Darstellung des Spielfeldes zu tun haben. So kann zu einem Punkt auf der Bühne die entsprechende Rasterposition ermittelt werden und umgekehrt. Außerdem wird die Erreichbarkeit eines Feldes und die Verschiebbarkeit einer Kiste berechnet. Eine rekursive Methode findet den jeweils kürzesten Weg zu einem angeklickten Feld.

Das Sokoban-Objekt repräsentiert die Lagerarbeiterfigur. Es zeichnet die Figur und bewegt sie gehend oder schiebend auf ein angegebenes Feld. Das Objekt kann außerdem aufgefordert werden, die aktuelle Position in einer Liste zu merken. Durch die undo()-Methode lässt sich der jeweils letzte Schritt rückgängig machen. Die reset()-Methode setzt die Figur auf den Anfangszustand zurück.

Das Kiste-Objekt ähnelt dem Sokoban-Objekt. Es fehlt ihm jedoch die gehen()- und schieben()-Methode. Statt​dessen wird die bewegen()-Methode von der schieben()-Methode des Sokoban-Objekts aufgerufen, um die Kiste zu bewegen. Das Kiste-Objekt implementiert ebenfalls die gleichen Methoden wie das Sokoban-Objekt, um eine History seiner Positionen zu erstellen und die letzten Schritte rückgängig zu machen. Zusätzlich bietet es mit der Methode istImZiel() die Möglichkeit abzufragen, ob es sich über einem Zielfeld befindet. Die Kisten-Objekte werden nicht in einer Globalen gespeichert, sondern sind nur über das KistenVerwalter-Objekt zu​gänglich.

Das KistenVerwalter-Objekt ist die einzige Möglichkeit, eine Kiste zu erzeugen und auf sie zuzugreifen. Mit ihm kann überprüft werden, ob an einer bestimmten Position auf dem Spielfeld eine Kiste steht und sich gege​benenfalls eine Referenz auf diese Liste zurückgeben lassen. Ansonsten delegiert es die zeichnen()-Methode sowie die Methoden zur Verwaltung der History an die einzelnen Kisten weiter. Die Methode sindKistenImZiel() greift auf die Methode istImZiel() aller Kisten zurück um festzustellen, ob alle Kisten auf einem Zielfeld stehen.

Das Spielstand-Objekt speichert jeweils ein Spielfeld-, Sokoban- und Kistenverwalter-Objekt, um bei einer Rückkehr zu einem Level den letzten Spielstand wieder herzustellen. Es stellt eine reine Datenstruktur dar, die implementierten Methoden sind lediglich Zugriffsmethoden auf die Objekteigenschaften.

Das Cursor-Objekt implementiert bis auf den new-Handler keine weiteren Methoden. Es ist lediglich ein Sammelobjekt für die Cursor-Konstanten, ein sogenanntes Utility-Objekt. Durch dieses Objekt kann auf sämt​liche benutzen Cursor über eine Globale zurückgegriffen werden, was den Programmcode übersichtlicher macht.

Objektverhältnisse des Sokoban-Spiels

5.14.3 Besonderheiten

So simpel die Idee hinter dem Sokoban-Spiel ist: in einem Spiel mit einem hohen Grad an Interaktivität bestimmen viele Parameter den Spielablauf. Entsprechend sinnvoll ist es, diese in logischen Objekten zu kap​seln. Zwar hätte man dieses Spiel auch mit nicht-objektorientierten Mitteln und einer ganzen Reihe von Listen programmieren können, jedoch wäre der Code schnell unübersichtlich geworden. So hätte man z.B. die Posi​tionen der Kisten in einer Liste speichern müssen, deren Spritenummer u.s.w. Außerdem hätte jede Änderung im Programm Auswirkungen auf andere Programmteile. Mit Objekten sind Änderungen leichter durchzu​führen. Wenn z.B. der Sokoban durch eine kleine Animation einer laufenden Spielfigur dargestellt werden soll, müssen nur die gehen- und schieben-Methoden des Sokoban-Objektes geändert werden.

Dennoch macht auch dieses Programm Gebrauch von herkömmlichen Programmiertechniken. Im Spielfeld​objekt wird beispielsweise rekursiv der kürzeste Weg berechnet, den der Sokoban benötigt, um zu einem be​stimmten Feld zu kommen. Die Methode _kuerzesterWeg(), die dafür benötigt wird, wird von keinem anderen Objekt als dem Spielfeldobjekt selbst gebraucht. Da in Lingo Methoden nicht privat deklariert werden können, wurde diese Methode mit einem Unterstrich gekennzeichnet. Als praktisch erweist sich bei dieser Methode das Lingo-Objekt Point, das zwei Integer- oder Floatwerte speichern kann. Die effizienteste Möglichkeit zum Auslesen der einzelnen Werte sind die Objekteigenschaften locH und locV. Diese Herangehensweise ist etwas schneller als das Erzeugen einer Liste mit zwei Einträgen und das Auslesen mit den „[ ]“-Operatoren.

Durch die Benutzung des Spielstandobjekts stellt sich das Speichern des Spielfeld-, Sokoban- und Kistenver​walterobjektes sehr einfach dar. Diese werden einfach dem new-Handler des Spielstandobjektes übergeben und die entsprechenden globalen Variablen mit neu instantiierten Objekten überschrieben. Dadurch ist es nicht nötig, eine Kopie der Objekte zu erstellen. Da die Kistenobjekte nur vom Kistenverwalter abhängig sind, wer​den diese automatisch mitgespeichert. Um den Spielstand wieder herzustellen, genügt es, die gespeicherten Werte den Globalen wieder zuzuordnen.

5.15 Tic Tac Toe

5.15.1 Beschreibung

Das TicTacToe-Spiel basiert auf dem Käsekästchenspiel, bei dem zwei Spieler versuchen, auf einem quadra​tischen, in Kästchen unterteilten Spielfeld, durch das Setzen von Symbolen eine Reihe zu bilden. Das Spiel kann wahlweise gegen den Computer gespielt werden, der Computer kann auch alleine gegen sich selbst spie​len. Die Größe des Spielfeldes kann vom klassischen 3×3 Spielfeld bis zu einer Größe von 7×7 Feldern variiert werden, ebenso kann die Zahl der Symbole, die für eine Reihe benötigt werden, zwischen 3 und 7 Felder einge​stellt werden. Bei einem größeren Spielfeldes ist es sinnvoll, nicht bis zur vollständigen Belegung aller Felder zu spielen, sondern das Spiel vorzeitig abzubrechen, wenn kein Spieler mehr in der Lage ist, eine Reihe zu bilden. Diese Option kann über ein Auswahlfeld angewählt werden.

Screenshot des Tic Tac Toe - Spieles

5.15.2 Definition der Objekte

Das Spielfeld-Objekt ist die logische Repräsentation des Spielfeldes. Sein new-Handler erhält als einziges Argument ein Spielfeldraster (eine verschachtelte Liste), das sozusagen vom Spielfeldobjekt „eingewickelt“ wird. Es bietet Methoden für das Auslesen und Setzen von Feldern und das Bestimmen der Größe des Rasters. Außerdem bietet es Methoden, um die Zahl der belegten und freien Felder zu bestimmen und um zu ermitteln, welcher Spieler gewonnen hat und welche Felder eine Lösungsreihe bilden. Zusätzlich gibt es einige Methoden, die vom Computergegner zur Ermittlung des besten Zuges benötigt werden. Diese ermitteln, wie viele Reihen unter bestimmten Bedingungen gebildet werden können. Damit die Gesamtzahl an möglichen Reihen nicht ständig neu berechnet werden muss, werden die möglichen Reihen pro Feld in der Eigenschaft mMoeglicheReihen festgehalten und die Ergebnisse wiederbenutzt. Beim Verändern des Spielfeldes wird diese Eigenschaft zurückgesetzt.

Um eine Kopie des Spielfeldes zu erstellen, bietet das Spielfeld-Objekt die Methode getCopy(). Die Kopie des Spielfeldes kann verändert werden, ohne dass sich die Originalwerte ändern. Der new-Handler des Spielfeld-Objektes wird übrigens nie direkt, sondern entweder beim erstellen einer Kopie oder durch das Objekt ZeichenbaresSpielfeldObjekt aufgerufen.

Das Objekt ZeichenbaresSpielfeldObjekt erweitert das Spielfeld-Objekt um die Eigenschaft, das Spielfeld auf dem Bildschirm darzustellen; es erbt alle Eigenschaften des Spielfeldes und fügt Methoden zum Zeichnen des Spielfeldes, zum Hervorheben der Gewinnerreihen und zum Aufblinkenlassen des letzten Computerzuges. Ferner speichert es die tatsächliche Größe des Spielfeldes auf dem Bildschirm und kann dadurch aus der Maus​position den entsprechenden Rasterindex des Spielfeldes berechnen.

Das Schiedsrichter-Objekt startet ein Spiel, hält fest, welcher Spieler am Zug ist und ob ein Spieler ziehen darf. Nach jedem Spielzug wird die Methode zugBeendet() aufgerufen, die prüft, ob ein Spieler gewonnen hat oder das Spiel unentschieden ausgegangen ist. Das Schiedsrichterobjekt fügt sich außerdem der actorlist hinzu und entscheidet im stepFrame-Handler, ob der Computer am Zug ist und führt gegebenenfalls dessen ziehen()-Methode aus.

Das Computergegner-Objekt bietet als einzige öffentliche Methode die Methode ziehen(), die den nächsten Zug ausführt. Diese führt daraufhin Berechnungen aus, um den bestmöglichen Zug zu ermitteln. Dabei wird nach einer bestimmten Reihenfolge jede freie Feldposition auf dem Spielfeld einer Wertung zugeordnet. In mehreren Berechnungsschritten werden jeweils die höchsten Wertungen weiter untersucht, bis am Ende eine Menge gleichwertiger Züge übrigbleibt. Aus dieser wird zufällig ein Zug ausgewählt ausgeführt. Leider war es aus Geschwindigkeitsgründen nicht möglich, den Computergegner einige Züge im Voraus denken zu lassen, da die zu bearbeitende Datenmenge sehr schnell ansteigt und Lingo wenig Möglichkeiten zur Optimierung des Codes lässt. Der Computer spielt also kein perfektes Spiel. Die genaue Vorgehensweise ist dem Lingo-Code zu entnehmen, der gut kommentiert ist.

Es wäre theoretisch möglich, vom Computergegner zwei Instanzen zu erzeugen, wenn der Computer gegen sich selbst spielt. Für unser Spiel ist dies jedoch nicht notwendig, da im Computergegner-Objekt keine über mehrere Züge relevanten Eigenschaften gespeichert werden. Dies kann in anderen Spielen durchaus der Fall sein, z.B. wenn der Computergegner eine bestimmte Taktik verfolgt. Bei einem Kartenspiel beispielsweise wird der Computer für jeden Spieler, je nach eigenem Blatt und ausgespielten Karten, eine eigene Wahrscheinlich​keit für bestimmte Karten von einem verdeckten Stapel berechnen. Der Vorteil an der objektorientierten Pro​grammierung ist, dass das Verhalten eines Spielers nur einmal programmiert werden muss und für jeden neuen Spieler einfach eine Instanz erzeugt wird.

Das Objekt SpielsituationsObjekt wird nur vom Computergegner benutzt. Es wird für eine Was-Wäre-Wenn-Analyse benutzt. Sein new-Handler erhält Informationen über das zu untersuchende Spielfeld, den aktuellen Spieler und den zu untersuchenden Zug. Dieser wird auf einer Kopie des Spielfelds ausgeführt. Anschließend kann untersucht werden, ob der Zug zum Gewinn eines Spielers führt (spielerGewinnnt())oder den Gegner vom Gewinnen abhält (gegnerGewinntNicht()). Des weiteren bietet das Objekt noch einige Methoden, um eine Wertung für die neue Situation zu berechnen.

Das Utilities-Objekt enthält zwei Methoden, die von mehreren Objekten verwendet werden, nämlich die Er​mittlung des Gegenspielers zu einem Spieler, und ob eine Zahl gerade oder ungerade ist. Die Objekte definieren das Utilities-Objekt als ancestor und können fortan diese Methoden benutzen, als wären es die eigenen.

Sequenzdiagramm des TicTacToe-Spiels

5.15.3 Besonderheiten

Dieses Spiel zeigt, dass sowohl der Computergegner wie auch der menschliche Spieler zum Setzen der Symbole auf dieselbe Methode zugreifen (gSpielfeldObj.setAtPos(indexPos, spieler)). Daher ist es sehr einfach, zwischen Computergegner und menschliche Gegner umzuschalten. Es wäre auch denkbar, dass das Spiel über ein Netzwerk gespielt wird und die Parteien sich über einen Multiuser-Server austauschen, an welche Stelle des Spielfeldes ein Symbol gesetzt werden soll.

Das Spiel zeigt ein Beispiel für die Kopie eines Objektes. Das Spielfeld wird vom Computergegner kopiert, um darauf alle Szenarien durchzuspielen, die entstehen, wenn ein Symbol auf ein freies Feld gesetzt wird. Dabei wird immer nur die logische Struktur des Spielfeldes kopiert, nicht die Eigenschaften, die das zeichenbare Spielfeldobjekt zur Darstellung benötigt. Durch diese Trennung wird der Speicherbedarf minimiert, da stets nur eine zeichenbare Instanz des Spielfeldes existiert. Durch das Ableiten eines Spielfeld-Objektes kann das zei​chenbare Spielfeld dennoch auf dessen Methoden zugreifen, als wären es seine eigenen.

Die Ausführung der ziehen()-Methode des Computergegners aus dem stepFrame-Handler heraus hat folgen​den Grund: Director prüft Benutzerereignisse wie z.B. Mausklicks immer nur ab, wenn der Abspielkopf in einen neuen Frame wechselt. Dies passiert nur, wenn alle Skripte eines Frames abgearbeitet sind. Solange einer der Spieler menschlich ist, ist dies auch der Fall, da der Rechner nach dem Computerzug auf eine Benutzerein​gabe wartet und sich in einer Schleife auf einem Frame befindet. Wenn jedoch beide Spieler vom Computer übernommen werden, würde ein Aufruf der ziehen()-Methode in der zugBeendet()-Methode des Schiedsrichter mit einem erneuten Aufruf von zugBeendet()-Methode abschließen. Diese würde wieder ziehen() aufrufen, u.s.w. Diese Verschachtelung an Methodenaufrufen würde Benutzereingaben blockieren, bis das Spiel beendet ist. Ein Aufruf der ziehen()-Methode aus einem stepFrame-Handler ist hingegen unproblematisch, da die Ausführung dieses Handlers voraussetzt, das ein Frame-Wechsel stattgefunden hat und somit Mausereignisse ausgewertet wurden.

In diesem Beispiel wurde für die Umsetzung gleichartiger Benutzerelemente Gebrauch von eigenen Verhalten und denen aus der Director-Verhaltensbibliothek gemacht. Die 4 Radio-Buttons zur Spielertyp-Auswahl benut​zen alle das gleiche Verhalten mit unterschiedlichen Eigenschaften für die Spielernummer und den Spielertyp. Ebenso basiert die Funktion der Schieberegler auf demselben Verhalten. Für die eingeblendeten Hilfefenster wurde das Tooltip-Verhalten und das Display-Text-Verhalten aus der Library verwendet, letzteres in einer leicht modifizierten Form.

6 Anhang: Neue Befehle in Director 8

6.1 handler()
Syntax:

Skriptobjekt.handler(#Handlername)

Gibt TRUE zurück, wenn das Skriptobjekt (Parent-Skript, Child-Objekt oder Verhalten) den angegebenen Handler enthält.

6.2 handlers()

Syntax:

Skriptobjekt.handlers()

Gibt eine lineare Liste zurück, die alle Handlernamen des Skriptobjekts als Symbol enthalten.

6.3 on isOKToAttach
Syntax:

on isOkayToAttach me, Skripttyp, Spritenummer

Mit diesem Handler ist es möglich zu verhindern, dass Verhalten einem falschen Darstellertypen zugeordnet wird oder dass Verhalten, die für den Skriptkanal bestimmt sind, mit einem Sprite verbunden werden und um​gekehrt. Er wird dem Verhalten hinzugefügt und automatisch aufgerufen, wenn ein Verhalten in das Drehbuch gezogen wird. Wenn der Rückgabewert TRUE ist, wird das Verhalten zugeordnet, ansonsten verweigert Director die Zuordnung.

Skripttyp kann zwei Werte haben: #graphic und #script; #script bedeutet, das Verhalten wird gerade in den Skriptkanal gezogen; #graphic bedeutet, es wird auf ein Sprite gezogen; im letzten Fall kann man mit dem Argument Spritenummer herausfinden, welchen Typs der Darsteller in diesem Spritekanal ist und ent​sprechend entscheiden, ob das Verhalten mit diesem Typ verbunden werden kann oder nicht.

6.4 rawNew()
Syntax:

Parentskript.rawNew()

Erzeugt eine Instanz des Parentskripts ohne den new-Handler aufzurufen. Um das Objekt zu initialisieren, kann der new-Handler später aufgerufen werden.

6.5 setScriptList()

Syntax:

sprite(Spritenummer).setScriptList(Skriptliste)

Diese Funktion dient dazu, die Initialisierungswerte eines Verhaltens per Lingo zu setzen, also die Eigen​schaften, die man im Verhaltensinspektor sieht. Die übergebene Skriptliste hat die gleiche Form wie die Liste, die die Eigenschaft the scriptList of sprite zurückgibt.

[[(Verhaltensdarsteller),"[#Eigenschaft1: Wert, #Eigenschaft2: Wert, ...]",

[(Verhaltensdarsteller),"[#Eigenschaft1: Wert, #Eigenschaft2: Wert, ...]"]]

Die Änderungen sind permanent.

Glossar

Abstraktion

Abstraktion ist eine Methode, bei der unter einem bestimmten Gesichtspunkt die wesentlichen Merk​male eines Gegenstandes oder Begriffes herausgesondert werden. ([Oestereich98], S.339)

Argument

Konkreter Wert eines Parameters. ([Oestereich98], S.341)

Datenabstraktion

Hierunter versteht man das Prinzip, nur die auf ein Objekt anwendbaren Operationen nach außen sichtbar zu machen. Die tatsächliche innere Realisierung der Operationen und die innere Struktur des Objektes werden verborgen, d.h. man betrachtet abstrakt die eigentliche Semantik und lässt die tat​sächliche Implementierung außer acht. ([Oestereich98], S.343)

In Lingo lässt sich dieses Prinzip zwar bewusst anwenden, es kann jedoch nicht verhindert werden, dass es unterwandert wird, da generell alle Methoden und Eigenschaften von außen zugänglich sind.

Dynamische Bindung, Späte Bindung

Hierunter ist zu verstehen, dass eine Nachricht erst zur Programmlaufzeit einer konkreten Operation zugeordnet wird, die diese Nachricht dann interpretiert. ([Oestereich98], S.344)

Der Methodenaufruf in Lingo basiert auf diesem Prinzip.

Garbage Collection

Die Garbage Collection sorgt dafür, dass die Laufzeitbibliothek, und nicht der Programmierer, den Speicherplatz von Objekten freigibt, die nicht mehr in Gebrauch sind.([Linden99], S. 12)

In Lingo wird für jede gespeicherte Objektreferenz ein interner Zähler hochgezählt. Verweist keine Variable mehr auf ein Objekt, d.h. ist der Zähler 0, wird der Speicherbereich freigegeben.

Klasse

Eine Klasse ist die Definition der Attribute, Operationen und der Semantik für eine Menge von Objekten. Alle Objekte einer Klasse entsprechen dieser Definition.([Oestereich98], S.347)

In Lingo wird nicht von Klassen gesprochen, sondern von parent scripts oder übergeordneten Skripten.

Nachricht

Nachrichten sind ein Mechanismus, mit dem Objekte untereinander kommunizieren können. Eine Nachricht überbringt einem Objekt die Information darüber, welche Aktivität von ihm erwartet wird, d.h. eine Nachricht fordert ein Objekt zur Ausführung einer Operation auf. Eine Nachricht besteht aus einem Selektor (einem Namen), einer Liste von Argumenten und geht an genau einen Empfänger. Der Sender einer Nachricht enthält ggf. ein Antwort-Objekt zurück. Durch Polymorphismus kann eine Nachricht zum Aufruf einer von mehreren gleichlautenden Operationen führen. ([Oestereich98], S.349)

Objekt

Ein Objekt ist eine konkret vorhandene und agierende Einheit mit eigener Identität und definierten Grenzen, das Zustand und Verhalten kapselt. Der Zustand wird repräsentiert durch die Attribute und Beziehungen, das Verhalten durch Operationen bzw. Methoden. Jedes Objekt ist Exemplar (Synonym: Instanz) einer Klasse. Das definierte Verhalten gilt für alle Objekte einer Klasse gleichermaßen, ebenso die Struktur ihrer Attribute. Die Werte der Attribute sind jedoch individuell für jedes Objekt. Jedes Objekt hat eine eigene, von seinen Attributen u.a. unabhängige, nicht veränderbare Identi​tät.([Oestereich98], S.350)

In Lingo heißen Objekte auch child objects oder untergeordnete Objekte.

Operation

Operationen sind Dienstleistungen, die von einem Objekt mit einer Nachricht angefordert werden können, um ein bestimmtes Verhalten zu bewirken. Sie werden implementiert durch Methoden. In der Praxis werden Operationen und Methoden häufig synonym verwendet. ([Oestereich98], S.351)

Parameter

Ein Parameter ist die Spezifikation einer Variablen, die Operationen, Nachrichten oder Ereignissen mitgegeben, von diesen verändert oder zurückgegeben wird. ([Oestereich98], S.352)

Polymorphismus

Polymorphismus (Vielgestaltigkeit) heißt, dass gleichlautende Nachrichten an kompatible Objekte unterschiedlicher Klassen ein unterschiedliches Verhalten bewirken können. Beim dynamischen Polymorphismus wird eine Nachricht nicht zur Kompilierzeit, sondern erst beim Empfang zur Pro​grammlaufzeit einer konkreten Operation zugeordnet. Voraussetzung hierfür ist das dynamische Bin​den. ([Oestereich98], S.352)

Lingo unterstützt dynamischen Polymorphismus. Der Nachteil des Prinzips ist jedoch, dass Pro​grammfehler erst zur Laufzeit entdeckt werden können.

Vererbung

Vererbung ist ein Programmiersprachenkonzept für die Umsetzung einer Relation zwischen einer Ober- und einer Unterklasse, wodurch Unterklassen die Eigenschaften ihrer Oberklassen mitbenutzen können. ([Oestereich98], S.357)

In Lingo lässt sich dies durch die Benutzung der ancestor-Eigenschaft erreichen. Von Ober- und Unterklassen kann jedoch nicht gesprochen werden, da der ancestor keine Klasse sondern ein Objekt ist, und man ihn während der Laufzeit ändern kann.

6.7 Zum Weiterlesen empfohlen

Peter Small

Lingo Sorcery Second Edition
John Wiley & Sons Ltd

Erschienen: 1999

550 Seiten

Die erste Ausgabe dieses Buch erschien bereits 1996. Daher wirken manche Passagen des Buches überholt, auch wenn die zweite Auflage auf den neuesten Stand gebracht wurde und jetzt in einem angehängten Kapitel Verhalten behandelt. Viele Vorgehensweisen muten dennoch überholt an. Hilfreich sind die zahlreichen Bei​spiele, in denen objektorientierte Programmierung Schritt für Schritt erklärt wird. Der Bereich Spiele bleibt jedoch außen vor; als einzige richtige Anwendung wird ein Avatar-System entwickelt. Die dafür verwendeten Techniken haben oft weniger mit OOP, sondern mit der Ausreizung einer Skriptsprache bis an ihre Grenzen zu tun. Peter Small verfolgt mit der Erzeugung von Objekten aus Textdateien, denen er mit dem do-Befehl Leben einhaucht, einen unkonventionell Ansatz. Trotzdem war das Buch nicht zu unrecht jahrelang der unangefoch​tene Spitzenreiter, da einziger Vertreter seiner Thematik.

Martin Kloss

Lingo objektorientiert - Director optimiert einsetzen
Galileo Press

Erschienen: 2000

528 Seiten

Dieses Buch geht das Thema Objektorientierung von der theoretischen Seite an. Wer konkrete Problem​lösungen oder Übungen erwartet, wird enttäuscht werden. Was das Buch jedoch bietet, ist eine kompetente Erklärung der objektorientierten Lingo-Programmierung und aller Befehle. Außerdem vernachlässigt Martin Kloss auch verwandte Thematiken, wie Listen, und programmiertechnische Details, wie die Übergabe von Parametern, nicht.

Gary Rosenzweig

Advanced Lingo for Games
Hayden Books

Erschienen: 2000

504 Seiten
Dieses Buch befasst sich mit der Spieleprogrammierung in Lingo wie kein anderes. Es geht gleich mit einer erschlagenden Fülle an praktischen und vollwertigen Beispielprogrammen zur Sache. Der Leser findet den kompletten Quellcode zu vielen Spielen aus fast allen Bereichen, welcher frei kopiert und verändert werden darf. Gary Rosenzweig ist im Multimedia-Bereich kein Unbekannter, seine Tipps und Lösungsvorschläge sind fundiert und bewährt. Vor allem Anfänger im Spielebereich profitieren von seinen nicht zu komplexen Bei​spielen. Hervorzuheben ist der rege Einsatz von Verhalten; auf klassisches Parent-Child-Scripting wird jedoch komplett verzichtet.

6.8 Empfehlenswerte Websites

Director Web (http://hakatai.mcli.dist.maricopa.edu/director/index.html)

Hervorragende Website mit einer erlesenen, thematisch geordneten Tipps & Tricks-Abteilung.

Direct-L (http://listserv.uark.edu/archives/direct-l.html)
Riesige,  aktuelle und durchsuchbare Newsgroup. Wenn ein Directorproblem schon einmal aufgetreten ist, findet man hier mit großer Wahrscheinlichkeit Informationen darüber.

UpdateStage (http://www.updatestage.com/)
Ausgesuchte Tips und die organisierteste Bugliste im Netz.

Macromedia (http://www.macromedia.com/)

Die offizielle Webseite mit einem großen Fundus an Technotes und fundamentierten Dokumentationen zu speziellen Problemen.

� [Dictionary7], S. 150


� [Epstein98], S. 342-343


� [Epstein98], S.343


� Intern benutzt Lingo ebenfalls das MUI-Xtra zur Darstellung der Dialogfenster.


� [Dictionary7], S. 76


� [Dictionary8] S. 224 ff.

